/*
 * Copyright 2011-2012 Rivoli team
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 * http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

package com.googlecode.rivoli.configuration

import com.googlecode.rivoli.RivoliException

import org.testng.annotations.Test

class ConfigurationTest {
	@Test
	void 'configurables should be remembered'() {
		Configuration configuration = new Configuration()
		Configurable testConfigurable = new TestConfigurable()

		configuration.addConfigurable('test', testConfigurable)

		assert configuration.getConfigurable('test') == testConfigurable
	}

	@Test(expectedExceptions=RivoliException)
	void 'adding duplicate configurable names should not be allowed'() {
		Configuration configuration = new Configuration()

		configuration.addConfigurable('test', new TestConfigurable())
		configuration.addConfigurable('test', new TestConfigurable())
	}

	@Test(expectedExceptions=RivoliException)
	void 'getting a configurable which does not exist should not be allowed'() {
		 new Configuration().getConfigurable('test')
	}

	@Test
	void 'configurables should be obtainable by type'() {
		Configuration configuration = new Configuration()

		configuration.addConfigurable('test', new TestConfigurable())
		configuration.addConfigurable('another', new AnotherConfigurable())

		assert configuration.getConfigurables(TestConfigurable).keySet() == ['test'] as Set
	}

	@Test
	void 'obtaining configurables of absent type should result in empty map'() {
		assert new Configuration().getConfigurables(TestConfigurable) == [:]
	}

	private class TestConfigurable implements Configurable {
		void init(Map<String, ?> parameters) {}
	}

	private class AnotherConfigurable implements Configurable {
		void init(Map<String, ?> parameters) {}
	}
}